Online Appendix to accompany Sterba, S.K. (In press). A latent transition analysis model for latent-state-dependent nonignorable missingness. *Psychometrika.*

Online Appendix Table of Contents:

M*plus* 7.11 Syntax for MNAR-PP LTA with missingness starting at time 2 (as in Equation (6) and Figure 2 Panel A) where *J*=8, *K*=4, *Q*=3, and *T*=3.

```
DATA:  FILE = yourdataset.dat;  ! identify dataset name

VARIABLE:
NAMES =  id t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t2m1-t2m8 t3m1-t3m8;
     ! list all variable names in the dataset
     ! t1y1-t1y8 are the J binary outcomes at time 1
     ! t2y1-t2y8 are the J binary outcomes at time 2
     ! t3y1-t3y8 are the J binary outcomes at time 3
     ! t2m1-t2m8 are the J binary missingness indicators at time 2
     ! t3m1-t3m8 are the J binary missingness indicators at time 3
MISSING= . ;  ! identify missingness code for y-outcomes in dataset (here, a period)
USEVARIABLES ARE t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t2m1-t2m8 t3m1-t3m8;
     ! identify all dataset variables used in this particular analysis
CATEGORICAL = t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t2m1-t2m8 t3m1-t3m8;
     ! declare all y-outcomes and missingness indicators as categorical
CLASSES = c1y (4) c2y (4) c2m (3) c3y (4) c3m (3) ;
     ! label t1, t2, t3 categorical latent variables in outcome process as c1y, c2y, c3y, respectively
     ! label t2, t3 categorical latent variables in missingness process as c2m, c3m, respectively
     ! In () specify # of latent states for categorical latent variables in outcome process (here 4)
     ! In () specify # of latent states for categorical latent variables in missingness process (here 3)
ANALYSIS:  TYPE = MIXTURE; STARTS=50 5; ESTIMATOR=ML;
     ! declare that model is a mixture and specify estimation options

MODEL:
%OVERALL%
     ! In %OVERALL% specify structural relations between outcome & missingness processes
c2y on c1y;          !corresponds with manuscript Equation (3)
c2m on c2y c1y;   !corresponds with manuscript Equation (7)
c3y on c2y;          !corresponds with manuscript Equation (3)
c3m on c2m c2y c3y; !corresponds with manuscript Equation (8)

     ! below, measurement invariance imposed within-state across-time in outcome process
     ! via list constraint (a1-a8) for thresholds in outcome state 1,
     ! and list constraint (b1-b8) for thresholds in outcome state 2,
     ! and list constraint (c1-c8) for thresholds in outcome state 3,
     ! and list constraint (d1-d8) for thresholds in outcome state 4

MODEL c1y: !specify outcome process submodel at time 1
%c1y#1%    !outcome state 1 at t1
[t1y1$1-t1y8$1] (a1-a8);   !thresholds for J=8 y-outcomes
%c1y#2%    !outcome state 2 at t1
[t1y1$1-t1y8$1] (b1-b8);  !thresholds for J=8 y-outcomes
%c1y#3%    !outcome state 3 at t1
[t1y1$1-t1y8$1] (c1-c8);   !thresholds for J=8 y-outcomes
%c1y#4%    !outcome state 4 at t1
[t1y1$1-t1y8$1] (d1-d8);  !thresholds for J=8 y-outcomes

MODEL c2y: !specify outcome process submodel at time 2
%c2y#1%    !outcome state 1 at t2
```

```
[t2y1$1-t2y8$1] (a1-a8);  !thresholds for J=8 y-outcomes
%c2y#2%   !outcome state 2 at t2
[t2y1$1-t2y8$1] (b1-b8); !thresholds for J=8 y-outcomes
%c2y#3%   !outcome state 3 at t2
[t2y1$1-t2y8$1] (c1-c8);  !thresholds for J=8 y-outcomes
%c2y#4%   !outcome state 4 at t2
[t2y1$1-t2y8$1] (d1-d8); !thresholds for J=8 y-outcomes

MODEL c3y:  !specify outcome process submodel at time 3
%c3y#1%   !outcome state 1 at t3
[t3y1$1-t3y8$1] (a1-a8);  !thresholds for J=8 y-outcomes
%c3y#2%   !outcome state 2 at t3
[t3y1$1-t3y8$1] (b1-b8); !thresholds for J=8 y-outcomes
%c3y#3%   !outcome state 3 at t3
[t3y1$1-t3y8$1] (c1-c8);  !thresholds for J=8 y-outcomes
%c3y#4%   !outcome state 4 at t3
[t3y1$1-t3y8$1] (d1-d8); !thresholds for J=8 y-outcomes

    ! below, measurement invariance imposed within-state across time in missingness process
    ! via list constraint (e1-e8) for thresholds in missingness state 1,
    ! and list constraint (f1-f8) for thresholds in missingness state 2,
    ! and list constraint (g1-g8) for thresholds in missingness state 3

MODEL c2m: !specify missingness process submodel at time 2
%c2m#1%   !missingness state 1 at t2
[t2m1$1-t2m8$1] (e1-e8); !thresholds for J=8 m-indicators
%c2m#2%   !missingness state 2 at t2
[t2m1$1-t2m8$1] (f1-f8);  !thresholds for J=8 m-indicators
%c2m#3%   !missingness state 3 at t2
[t2m1$1-t2m8$1] (g1-g8); !thresholds for J=8 m-indicators

MODEL c3m: !specify missingness process submodel at time 3
%c3m#1%   !missingness state 1 at t3
[t3m1$1-t3m8$1] (e1-e8); !thresholds for J=8 m-indicators
%c3m#2%   !missingness state 2 at t3
[t3m1$1-t3m8$1] (f1-f8);   !thresholds for J=8 m-indicators
%c3m#3%   !missingness state 3 at t3
[t3m1$1-t3m8$1] (g1-g8); !thresholds for J=8 m-indicators
```

Note. MNAR-PP LTA= Missing not at random parallel process latent transition model.
Missingness indicators t2m1-t2m8 t3m1-t3m8 are 1 if missing, 0 if present.

M*plus* 7.11 Syntax for MNAR-PP LTA with missingness starting at time 1 (as in Equation (11) and Figure 2 Panel B) where *J*=8, *K*=4, and *Q*=3

```
    !comments only provided for commands that differ from previous syntax
DATA:  FILE = yourdataset2.dat;
VARIABLE: NAMES =  id t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t1m1-t1m8 t2m1-t2m8 t3m1-t3m8;
    ! now dataset also contains t1m1-t1m8, the J binary missingness indicators at time 1
MISSING= . ;
USEVARIABLES ARE t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t1m1-t1m8 t2m1-t2m8 t3m1-t3m8;
CATEGORICAL = t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t1m1-t1m8 t2m1-t2m8 t3m1-t3m8;
    ! now t1m1-t1m8 are also used in analysis and declared as categorical
CLASSES = c1y (4) c1m (3) c2y (4) c2m (3) c3y (4) c3m (3) ;
    ! now also include label for t1 categorical latent variable in missingness process: c1m
ANALYSIS: TYPE = MIXTURE; STARTS=50 5; ESTIMATOR=ML;
MODEL:
%OVERALL% !now structural relations reflect Equation (11) and Figure 2 Panel B
c1m on c1y; !now t1 missingness states are regressed on t1 outcome states
c2y on c1y;
c2m on c2y c1y c1m; !now t2 missingness states are also regressed on t1 outcome states
c3y on c2y;
c3m on c2m c2y c3y;

MODEL c1y:
%c1y#1%  [t1y1$1-t1y8$1] (a1-a8);
%c1y#2%  [t1y1$1-t1y8$1] (b1-b8);
%c1y#3%  [t1y1$1-t1y8$1] (c1-c8);
%c1y#4%  [t1y1$1-t1y8$1] (d1-d8);
MODEL c2y:
%c2y#1%  [t2y1$1-t2y8$1] (a1-a8);
%c2y#2%  [t2y1$1-t2y8$1] (b1-b8);
%c2y#3%  [t2y1$1-t2y8$1] (c1-c8);
%c2y#4%  [t2y1$1-t2y8$1] (d1-d8);
MODEL c3y:
%c3y#1%  [t3y1$1-t3y8$1] (a1-a8);
%c3y#2%  [t3y1$1-t3y8$1] (b1-b8);
%c3y#3%  [t3y1$1-t3y8$1] (c1-c8);
%c3y#4%  [t3y1$1-t3y8$1] (d1-d8);

    !below, measurement invariance imposed within-state across times 1-3 in missingness process
MODEL c1m: !specify missingness process submodel at time 1
%c1m#1%    !missingness state 1 at t1
[t1m1$1-t1m8$1] (e1-e8); !thresholds for J=8 m-indicators
%c1m#2%    !missingness state 2 at t1
[t1m1$1-t1m8$1] (f1-f8);  !thresholds for J=8 m-indicators
%c1m#3%    !missingness state 3 at t1
[t1m1$1-t1m8$1] (g1-g8); !thresholds for J=8 m-indicators
MODEL c2m:
%c2m#1%  [t2m1$1-t2m8$1] (e1-e8);
%c2m#2%  [t2m1$1-t2m8$1] (f1-f8);
%c2m#3%  [t2m1$1-t2m8$1] (g1-g8);
MODEL c3m:
```

```
%c3m#1%  [t3m1$1-t3m8$1] (e1-e8);
%c3m#2%  [t3m1$1-t3m8$1] (f1-f8);
%c3m#3%  [t3m1$1-t3m8$1] (g1-g8);
```

Note. MNAR-PP LTA= Missing not at random parallel process latent transition model. Missingness indicators t1m1-t1m8, t2m1-t2m8, and t3m1-t3m8 are 1 if missing, 0 if present.

Online Appendix

Online Appendix Table OA1. Simulation results for *N*=1000: Percent Absolute Relative Bias (%ARB) for multinomial coefficient structural parameters in the outcome process

Parameter	Pop. Value	MNAR Missingness mechanism				MAR Missingness mechanism			
		Fit MNAR-PP LTA		Fit Conventional LTA		Fit MNAR-PP LTA		Fit Conventional LTA	
		Avg Est	%ARB	Avg Est	%ARB	Avg Est	%ARB	Avg Est	%ARB
$\omega_{k_1=1}$	-1.2	-1.213	1.10	-1.231	2.56	-1.212	1.02	-1.214	1.20
$\omega_{k_1=2}$	-0.9	-0.874	2.90	-0.831	7.67	-0.837	6.95	-0.848	5.73
$\alpha_{k_2=1}$	-1.2	-1.202	0.20	-1.305	8.74	-1.238	3.17	-1.235	2.88
$\alpha_{k_2=2}$	-0.9	-0.900	0.05	-0.894	0.66	-0.862	4.23	-0.905	0.52
$\alpha_{k_3=1}$	-1.2	-1.247	3.90	-1.411	17.57	-1.250	4.14	-1.253	4.42
$\alpha_{k_3=2}$	-0.9	-0.942	4.63	-0.953	5.89	-0.949	5.42	-0.940	4.42
$\beta_{k_2=1\|c_1^y=1}$	2.0	2.033	1.64	1.900	4.99	2.142	7.12	2.106	5.31
$\beta_{k_2=1\|c_1^y=2}$	1.0	0.956	4.45	0.760	23.97	1.008	0.77	1.032	3.15
$\beta_{k_2=2\|c_1^y=1}$	1.0	0.873	12.74	0.722	27.82	1.048	4.78	1.016	1.62
$\beta_{k_2=2\|c_1^y=2}$	1.5	1.530	2.01	1.256	16.28	1.543	2.84	1.593	6.23
$\beta_{k_3=1\|c_2^y=1}$	1.75	1.815	3.73	1.762	0.70	1.861	6.34	1.849	5.68
$\beta_{k_3=1\|c_2^y=2}$	1.0	1.056	5.58	0.798	20.24	1.037	3.72	1.058	5.84
$\beta_{k_3=2\|c_2^y=1}$	1.0	0.958	4.15	0.769	23.06	1.042	4.16	0.999	0.09
$\beta_{k_3=2\|c_2^y=2}$	1.5	1.636	9.05	1.430	4.65	1.682	12.12	1.682	12.14
Average %ARB			**4.34**		**12.88**		**4.90**		**4.36**

Notes. MNAR-PP LTA=Missing-not-at-random parallel process LTA; Pop. value=population parameter value. MAR=missing-at-random. Avg. Est.=average estimate. Tabled estimates are from samples that converged without estimation problems. Specifically, for samples generated with MAR missingness, 338-402 encountered no estimation problems, depending on fitted model. For samples generated with MNAR missingness, 291-302 encountered no estimation problems, depending on fitted model. Average %ARB is computed for the multinomial coefficient structural parameters pertaining to timepoints ≥ 2 (there was no missingness at *t*=1).

Online Appendix Table OA2. Simulation results for *N*=1000: Percent Absolute Relative Bias (%ARB) for threshold measurement parameters in the outcome process

Parameter	Pop. Value	MNAR Missingness mechanism				MAR Missingness mechanism			
		Fit MNAR-PP LTA		Fit Conventional LTA		Fit MNAR-PP LTA		Fit Conventional LTA	
		Avg Est	%ARB	Avg Est	%ARB	Avg Est	%ARB	Avg Est	%ARB
$v_{y1t\|k_t=1}$	-0.90	-0.900	0.05	-0.941	4.52	-0.922	2.47	-0.917	1.92
$v_{y2t\|k_t=1}$	-1.69	-1.819	7.64	-1.894	12.10	-1.814	7.32	-1.826	8.03
$v_{y3t\|k_t=1}$	-2.20	-2.314	5.20	-2.363	7.39	-2.293	4.24	-2.293	4.22
$v_{y4t\|k_t=1}$	-1.25	-1.267	1.39	-1.330	6.41	-1.277	2.20	-1.279	2.30
$v_{y5t\|k_t=1}$	-1.48	-1.552	4.87	-1.657	11.94	-1.566	5.81	-1.555	5.05
$v_{y1t\|k_t=2}$	0.32	0.310	3.07	0.307	4.05	0.338	5.51	0.326	1.97
$v_{y2t\|k_t=2}$	0.95	0.971	2.18	0.965	1.54	0.981	3.29	0.978	2.93
$v_{y3t\|k_t=2}$	-0.20	-0.189	5.43	-0.174	13.13	-0.192	4.07	-0.193	3.45
$v_{y4t\|k_t=2}$	0.15	0.157	4.89	0.147	2.28	0.152	1.57	0.149	0.90
$v_{y5t\|k_t=2}$	0.55	0.573	4.12	0.571	3.84	0.573	4.11	0.562	2.21
$v_{y1t\|k_t=3}$	2.10	2.124	1.16	2.152	2.47	2.145	2.16	2.139	1.85
$v_{y2t\|k_t=3}$	2.67	2.687	0.63	2.663	0.27	2.710	1.50	2.704	1.28
$v_{y3t\|k_t=3}$	1.80	1.834	1.90	1.849	2.72	1.859	3.27	1.838	2.11
$v_{y4t\|k_t=3}$	2.55	2.611	2.39	2.668	4.64	2.688	5.42	2.648	3.84
$v_{y5t\|k_t=3}$	2.30	2.323	1.01	2.327	1.16	2.346	1.99	2.339	1.69
Average %ARB			**3.06**		**5.23**		**3.66**		**2.92**

Notes. See Table OA.1 notes.

Online Appendix Table OA3. Structural parameter results from Manuscript Table 2 converted into a probability metric using Equations (2)-(4): Marginal and Conditional probabilities from the outcome process

Parameter	Pop. Value	MNAR Missingness mechanism		MAR Missingness mechanism	
		Fit MNAR-PP LTA	Fit Conventional LTA	Fit MNAR-PP LTA	Fit Conventional LTA
		Avg Est	Avg Est	Avg Est	Avg Est
$\pi_{k_1=1}$	.176	.176	.170	.176	.175
$\pi_{k_1=2}$	.238	.241	.240	.240	.240
$\pi_{k_1=3}$	.586	.583	.591	.584	.585
$\pi_{k_2=1}$	.247	.248	.225	.246	.246
$\pi_{k_2=2}$	.304	.304	.284	.308	.308
$\pi_{k_2=3}$	.449	.448	.491	.446	.446
$\pi_{k_3=1}$	.259	.259	.222	.259	.258
$\pi_{k_3=2}$	.330	.332	.308	.332	.333
$\pi_{k_3=3}$	.411	.409	.470	.410	.409
$\tau_{k_2=1\|k_1=1}$	.514	.520	.483	.515	.515
$\tau_{k_2=1\|k_1=2}$	.225	.223	.196	.222	.221
$\tau_{k_2=1\|k_1=3}$	.176	.176	.163	.175	.175
$\tau_{k_2=2\|k_1=1}$	.255	.250	.224	.256	.256
$\tau_{k_2=2\|k_1=2}$	.500	.504	.465	.506	.508
$\tau_{k_2=2\|k_1=3}$	.238	.238	.228	.242	.242
$\tau_{k_2=3\|k_1=1}$	.231	.230	.293	.229	.229
$\tau_{k_2=3\|k_1=2}$	.275	.272	.339	.272	.271
$\tau_{k_2=3\|k_1=3}$	.586	.586	.609	.583	.583
$\tau_{k_3=1\|k_2=1}$	.452	.454	.420	.455	.454
$\tau_{k_3=1\|k_2=2}$	.225	.223	.184	.224	.223
$\tau_{k_3=1\|k_2=3}$	.176	.176	.154	.175	.174
$\tau_{k_3=2\|k_2=1}$	.288	.287	.261	.288	.288
$\tau_{k_3=2\|k_2=2}$	.500	.504	.474	.502	.504
$\tau_{k_3=2\|k_2=3}$	.238	.239	.233	.238	.239
$\tau_{k_3=3\|k_2=1}$	.261	.258	.319	.258	.258
$\tau_{k_3=3\|k_2=2}$	.275	.273	.342	.274	.273
$\tau_{k_3=3\|k_2=3}$	.586	.585	.613	.587	.587

Notes. Pop. value=population parameter value. Avg. Est.=average estimate.

Online Appendix Table OA4. Measurement parameter results from Manuscript Table 3 converted into a probability metric using Equation (5): Item endorsement probabilities from the outcome process

Parameter	Pop. Value	MNAR Missingness mechanism		MAR Missingness mechanism	
		Fit MNAR-PP LTA	Fit Conventional LTA	Fit MNAR-PP LTA	Fit Conventional LTA
		Avg Est	Avg Est	Avg Est	Avg Est
$\rho_{y1t\|k_t=1}$	.711	.712	.720	.712	.712
$\rho_{y2t\|k_t=1}$	.844	.847	.856	.847	.848
$\rho_{y3t\|k_t=1}$	.900	.902	.907	.901	.901
$\rho_{y4t\|k_t=1}$	.777	.776	.785	.778	.778
$\rho_{y5t\|k_t=1}$	.815	.816	.824	.816	.816
$\rho_{y1t\|k_t=2}$	.421	.418	.427	.419	.419
$\rho_{y2t\|k_t=2}$	.279	.277	.290	.277	.277
$\rho_{y3t\|k_t=2}$	.550	.547	.556	.550	.550
$\rho_{y4t\|k_t=2}$	.463	.462	.474	.462	.462
$\rho_{y5t\|k_t=2}$	.366	.363	.375	.365	.365
$\rho_{y1t\|k_t=3}$	.109	.108	.111	.108	.108
$\rho_{y2t\|k_t=3}$	.065	.064	.066	.065	.065
$\rho_{y3t\|k_t=3}$	.142	.141	.144	.141	.141
$\rho_{y4t\|k_t=3}$	.072	.071	.073	.071	.071
$\rho_{y5t\|k_t=3}$	.091	.091	.093	.090	.090

Notes. LTA= latent transition analysis; MNAR-PP LTA=missing-not-at-random parallel process LTA; Pop. value=population parameter value. MAR=missing-at-random. Avg. Est.=average estimate.

Online Appendix Table OA5. Structural parameter results from Manuscript Table 4 converted into a probability metric using formulas in Equations (7) and (8): Selected conditional probabilities from the missingness process

	MNAR Missingness mechanism Fit MNAR-PP LTA		MAR Missingness mechanism Fit MNAR-PP LTA	
Parameter	Pop. Value	Avg Est	Pop. Value	Avg Est
$\tau_{q_2=1\|k_1=1,k_2=1}$	.777	.779	.254	.251
$\tau_{q_2=1\|k_1=2,k_2=1}$	.622	.624	.254	.251
$\tau_{q_2=1\|k_1=3,k_2=1}$	.182	.178	.254	.250
$\tau_{q_2=1\|k_1=1,k_2=2}$	.688	.678	.254	.248
$\tau_{q_2=1\|k_1=2,k_2=2}$	.500	.498	.254	.249
$\tau_{q_2=1\|k_1=3,k_2=2}$	.119	.114	.254	.248
$\tau_{q_2=1\|k_1=1,k_2=3}$	.223	.216	.254	.255
$\tau_{q_2=1\|k_1=2,k_2=3}$	.119	.115	.254	.255
$\tau_{q_2=1\|k_1=3,k_2=3}$	.018	.017	.254	.254
$\tau_{q_2=2\|k_1=1,k_2=1}$	.223	.221	.746	.749
$\tau_{q_2=2\|k_1=2,k_2=1}$	.378	.376	.746	.749
$\tau_{q_2=2\|k_1=3,k_2=1}$	.818	.822	.746	.750
$\tau_{q_2=2\|k_1=1,k_2=2}$	.312	.322	.746	.752
$\tau_{q_2=2\|k_1=2,k_2=2}$	.500	.502	.746	.751
$\tau_{q_2=2\|k_1=3,k_2=2}$	.881	.886	.746	.752
$\tau_{q_2=2\|k_1=1,k_2=3}$	.777	.784	.746	.745
$\tau_{q_2=2\|k_1=2,k_2=3}$	.881	.885	.746	.745
$\tau_{q_2=2\|k_1=3,k_2=3}$	.982	.983	.746	.746
$\tau_{q_3=1\|q_2=1,k_2=1,k_3=1}$	.940	.942	.508	.509
$\tau_{q_3=1\|q_2=2,k_2=1,k_3=1}$	.905	.906	.385	.385
$\tau_{q_3=1\|q_2=1,k_2=2,k_3=1}$	.852	.854	.508	.509
$\tau_{q_3=1\|q_2=2,k_2=2,k_3=1}$	.777	.778	.385	.384
$\tau_{q_3=1\|q_2=1,k_2=3,k_3=1}$	.500	.500	.508	.509
$\tau_{q_3=1\|q_2=2,k_2=3,k_3=1}$	.378	.374	.385	.384
$\tau_{q_3=1\|q_2=1,k_2=1,k_3=2}$	.852	.855	.508	.504
$\tau_{q_3=1\|q_2=2,k_2=1,k_3=2}$	.777	.779	.385	.379
$\tau_{q_3=1\|q_2=1,k_2=2,k_3=2}$	.679	.681	.508	.504

$\tau_{q_3=1\|q_2=2,k_2=2,k_3=2}$	.562	.561	.385	.379
$\tau_{q_3=1\|q_2=1,k_2=3,k_3=2}$	.269	.267	.508	.503
$\tau_{q_3=1\|q_2=2,k_2=3,k_3=2}$	.182	.179	.385	.379
$\tau_{q_3=1\|q_2=1,k_2=1,k_3=3}$	.500	.487	.508	.509
$\tau_{q_3=1\|q_2=2,k_2=1,k_3=3}$	.378	.362	.385	.385
$\tau_{q_3=1\|q_2=1,k_2=2,k_3=3}$	.269	.256	.508	.509
$\tau_{q_3=1\|q_2=2,k_2=2,k_3=3}$	.182	.171	.385	.384
$\tau_{q_3=1\|q_2=1,k_2=3,k_3=3}$	.060	.055	.508	.509
$\tau_{q_3=1\|q_2=2,k_2=3,k_3=3}$	.037	.034	.385	.384
$\tau_{q_3=2\|q_2=1,k_2=1,k_3=1}$	.060	.058	.493	.491
$\tau_{q_3=2\|q_2=2,k_2=1,k_3=1}$	.095	.094	.615	.615
$\tau_{q_3=2\|q_2=1,k_2=2,k_3=1}$	.148	.146	.493	.491
$\tau_{q_3=2\|q_2=2,k_2=2,k_3=1}$	.223	.222	.615	.616
$\tau_{q_3=2\|q_2=1,k_2=3,k_3=1}$	.500	.500	.493	.491
$\tau_{q_3=2\|q_2=2,k_2=3,k_3=1}$	.622	.626	.615	.616
$\tau_{q_3=2\|q_2=1,k_2=1,k_3=2}$	.148	.145	.493	.496
$\tau_{q_3=2\|q_2=2,k_2=1,k_3=2}$	.223	.221	.615	.621
$\tau_{q_3=2\|q_2=1,k_2=2,k_3=2}$	.321	.319	.493	.496
$\tau_{q_3=2\|q_2=2,k_2=2,k_3=2}$	.438	.439	.615	.621
$\tau_{q_3=2\|q_2=1,k_2=3,k_3=2}$	.731	.733	.493	.497
$\tau_{q_3=2\|q_2=2,k_2=3,k_3=2}$	.818	.821	.615	.621
$\tau_{q_3=2\|q_2=1,k_2=1,k_3=3}$	.500	.513	.493	.491
$\tau_{q_3=2\|q_2=2,k_2=1,k_3=3}$	.622	.638	.615	.615
$\tau_{q_3=2\|q_2=1,k_2=2,k_3=3}$	.731	.744	.493	.491
$\tau_{q_3=2\|q_2=2,k_2=2,k_3=3}$	.818	.829	.615	.616
$\tau_{q_3=2\|q_2=1,k_2=3,k_3=3}$	.940	.945	.493	.491
$\tau_{q_3=2\|q_2=2,k_2=3,k_3=3}$	.963	.966	.615	.616

Notes. LTA= latent transition analysis; MNAR-PP LTA=missing-not-at-random parallel process LTA; Pop. value=population parameter value. MAR=missing-at-random. Avg. Est.=average estimate.

Online Appendix Table OA6. Measurement parameter results from Manuscript Table 4 converted into a probability metric using Equation (9): Item endorsement probabilities from the missingness process

		MNAR Missingness mechanism	MAR Missingness mechanism
Parameter	Pop. Value	Fit MNAR-PP LTA	Fit MNAR-PP LTA
		Avg Est	Avg Est
$\rho_{m1t\|q_t=1}$	.681	.681	.682
$\rho_{m2t\|q_t=1}$	.613	.612	.612
$\rho_{m3t\|q_t=1}$	.715	.715	.715
$\rho_{m4t\|q_t=1}$	.646	.646	.645
$\rho_{m5t\|q_t=1}$	.657	.657	.658
$\rho_{m1t\|q_t=2}$	.165	.165	.165
$\rho_{m2t\|q_t=2}$	.095	.095	.095
$\rho_{m3t\|q_t=2}$	.146	.145	.146
$\rho_{m4t\|q_t=2}$	.076	.076	.076
$\rho_{m5t\|q_t=2}$	.119	.119	.119

Notes. LTA= latent transition analysis; MNAR-PP LTA=missing-not-at-random parallel process LTA; Pop. value=population parameter value. MAR=missing-at-random. Avg. Est.=average estimate.

Online Appendix Table OA7. Standard Errors for multinomial coefficient structural parameters in the outcome process

	MNAR Missingness mechanism						MAR Missingness mechanism					
	Fit MNAR-PP LTA			Fit Conventional LTA			Fit MNAR-PP LTA			Fit Conventional LTA		
	Avg *SE*	Empiri-cal SD	%ARB	Avg *SE*	Empiri-cal SD	%ARB	Avg *SE*	Empiri-cal SD	%ARB	Avg *SE*	Empiri-cal SD	%ARB
$SE(\omega_{k_1=1})$	.105	.102	2.86	.135	.128	4.53	.113	.107	5.51	.111	.107	3.48
$SE(\omega_{k_1=2})$	.097	.097	0.33	.123	.124	0.78	.118	.118	0.17	.115	.118	2.29
$SE(\alpha_{k_2=1})$	.120	.113	5.92	.145	.136	5.90	.121	.121	0.35	.118	.120	1.49
$SE(\alpha_{k_2=2})$	.120	.120	0.62	.145	.142	1.52	.148	.148	0.60	.135	.137	1.61
$SE(\alpha_{k_3=1})$	.123	.126	2.79	.153	.153	0.02	.147	.139	4.98	.143	.138	3.97
$SE(\alpha_{k_3=2})$	.158	.148	6.34	.181	.166	8.32	.185	.190	2.87	.174	.177	1.64
$SE(\beta_{k_2=1\|c_1^y=1})$	.161	.158	1.93	.165	.162	2.16	.160	.165	3.08	.158	.164	3.87
$SE(\beta_{k_2=1\|c_1^y=2})$	.240	.233	2.85	.259	.247	4.73	.238	.241	1.30	.237	.241	1.39
$SE(\beta_{k_2=2\|c_1^y=1})$	.293	.284	3.04	.347	.327	5.51	.275	.274	0.29	.275	.274	0.21
$SE(\beta_{k_2=2\|c_1^y=2})$	.255	.255	0.01	.270	.277	2.69	.286	.287	0.06	.288	.287	0.39
$SE(\beta_{k_3=1\|c_2^y=1})$	.172	.172	0.12	.166	.170	2.21	.173	.168	2.77	.169	.166	1.36
$SE(\beta_{k_3=1\|c_2^y=2})$	.260	.254	2.10	.305	.293	3.79	.257	.264	2.55	.251	.263	4.71
$SE(\beta_{k_3=2\|c_2^y=1})$	.317	.326	2.89	.323	.321	0.47	.276	.275	0.35	.274	.274	0.15
$SE(\beta_{k_3=2\|c_2^y=2})$	.312	.299	4.19	.354	.325	8.20	.340	.339	0.08	.336	.339	1.04
Average %ARB			**2.57**			**3.63**			**1.78**			**1.97**

Notes. Empirical SD is the standard deviation of the empirical sampling distribution of the parameter. MNAR-PP LTA=Missing-not-at-random parallel process LTA; Pop. value=population parameter value. MAR=missing-at-random. Avg. Est.=average estimate.

Online Appendix Table OA8. Standard Errors for threshold measurement parameters in the outcome process

	MNAR Missingness mechanism						MAR Missingness mechanism					
	Fit MNAR-PP LTA			Fit Conventional LTA			Fit MNAR-PP LTA			Fit Conventional LTA		
	Avg *SE*	Empiri-cal SD	%ARB	Avg *SE*	Empiri-cal SD	%ARB	Avg *SE*	Empiri-cal SD	%ARB	Avg *SE*	Empiri-cal SD	%ARB
$SE(v_{y1t\|k_t=1})$	.080	.078	1.71	.084	.084	0.10	.074	.071	3.69	.074	.071	3.41
$SE(v_{y2t\|k_t=1})$	.202	.192	4.95	.243	.229	5.92	.193	.177	8.30	.193	.177	8.08
$SE(v_{y3t\|k_t=1})$	.156	.161	3.01	.179	.179	0.12	.148	.143	3.16	.146	.143	2.15
$SE(v_{y4t\|k_t=1})$	.086	.089	2.43	.094	.096	1.63	.081	.081	0.81	.081	.081	0.28
$SE(v_{y5t\|k_t=1})$	.138	.129	6.57	.160	.146	8.73	.123	.118	3.55	.121	.118	2.10
$SE(v_{y1t\|k_t=2})$	.112	.104	6.55	.144	.132	8.32	.120	.113	6.14	.118	.113	4.50
$SE(v_{y2t\|k_t=2})$	.152	.145	4.60	.185	.179	3.23	.162	.156	4.01	.160	.156	2.48
$SE(v_{y3t\|k_t=2})$	.132	.125	4.94	.169	.161	4.72	.145	.136	6.01	.143	.136	4.89
$SE(v_{y4t\|k_t=2})$	.120	.118	1.62	.160	.152	4.90	.135	.129	4.76	.133	.129	3.19
$SE(v_{y5t\|k_t=2})$	.130	.124	4.68	.162	.156	3.33	.140	.134	4.57	.139	.134	3.68
$SE(v_{y1t\|k_t=3})$	.070	.070	0.39	.082	.084	1.77	.084	.085	1.18	.084	.085	1.56
$SE(v_{y2t\|k_t=3})$	.086	.082	4.80	.099	.093	5.97	.101	.095	5.50	.100	.095	5.04
$SE(v_{y3t\|k_t=3})$	.069	.068	1.17	.087	.084	3.38	.085	.085	0.57	.084	.085	1.68
$SE(v_{y4t\|k_t=3})$	.111	.107	3.30	.143	.138	3.07	.139	.137	1.06	.137	.137	0.30
$SE(v_{y5t\|k_t=3})$	.071	.071	0.10	.087	.083	4.65	.088	.086	2.17	.086	.086	0.43
Average %ARB			**3.39**			**3.99**			**3.70**			**2.92**

Notes. Empirical SD is the standard deviation of the empirical sampling distribution of the parameter. MNAR-PP LTA=Missing-not-at-random parallel process LTA; Pop. value=population parameter value. MAR=missing-at-random. Avg. Est.=average estimate.

Online Appendix Table OA9. Standard Errors for multinomial coefficient structural parameters and threshold measurement parameters in the missingness process (available only when MNAR-PP LTA is fit)

Parameter	MNAR Missingness Mechanism Fit MNAR-PP LTA			MAR Missingness Mechanism Fit MNAR-PP LTA		
	Avg. SE	Empirical SD	%ARB	Avg. SE	Empirical SD	%ARB
$SE(\alpha_{q_2=1})$	.419	.424	1.08	.098	.098	0.14
$SE(\alpha_{q_3=1})$	.341	.332	2.48	.097	.097	0.31
$SE(\beta_{q_2=1\|c_1^y=1})$	.197	.189	4.30	.131	.136	4.26
$SE(\beta_{q_2=1\|c_1^y=2})$	.228	.228	0.14	.202	.191	4.99
$SE(\beta_{q_2=1\|c_2^y=1})$	.380	.385	1.24	.160	.161	0.53
$SE(\beta_{q_2=1\|c_2^y=2})$	.502	.514	2.42	.327	.317	2.90
$SE(\beta_{q_3=1\|c_2^m=1})$	.149	.152	1.89	.087	.088	1.29
$SE(\beta_{q_3=1\|c_2^y=1})$	.211	.218	3.23	.116	.117	0.86
$SE(\beta_{q_3=1\|c_2^y=2})$	.239	.247	3.31	.174	.179	2.76
$SE(\beta_{q_3=1\|c_3^y=1})$	.309	.303	1.99	.134	.136	1.75
$SE(\beta_{q_3=1\|c_3^y=2})$	.466	.460	1.38	.274	.276	0.52
Average %ARB			**2.13**			**1.84**
$SE(v_{m1t\|q_t=1})$	.045	.042	5.87	.041	.045	10.95
$SE(v_{m2t\|q_t=1})$	.040	.040	1.88	.045	.043	4.25
$SE(v_{m3t\|q_t=1})$	.044	.045	2.42	.047	.048	2.10
$SE(v_{m4t\|q_t=1})$	.042	.042	0.62	.045	.045	1.72
$SE(v_{m5t\|q_t=1})$	.041	.042	2.47	.044	.045	0.89
$SE(v_{m1t\|q_t=2})$	.038	.037	4.27	.040	.039	4.01
$SE(v_{m2t\|q_t=2})$	.049	.048	3.02	.054	.051	5.28
$SE(v_{m3t\|q_t=2})$	.038	.040	3.57	.043	.042	1.62
$SE(v_{m4t\|q_t=2})$	.055	.055	0.91	.058	.060	2.88
$SE(v_{m5t\|q_t=2})$	.041	.043	4.87	.045	.046	2.08
Average %ARB			**2.99**			**3.58**

Notes. MNAR-PP LTA=missing-not-at-random parallel process LTA; Pop. value=population parameter value. MAR=missing-at-random. Avg. Est.=average estimate.

Table OA10. Simulation results: Coverage for multinomial coefficient structural parameters in the outcome process

Parameter	MNAR Missingness mechanism		MAR Missingness mechanism	
	Fit MNAR-PP LTA	Fit Conventional LTA	Fit MNAR-PP LTA	Fit Conventional LTA
$\omega_{k_1=1}$	.952	.940	.946	.938
$\omega_{k_1=2}$	.950	.948	.968	.970
$\alpha_{k_2=1}$	.940	.894	.952	.942
$\alpha_{k_2=2}$	.954	.898	.952	.954
$\alpha_{k_3=1}$	.956	.814	.948	.944
$\alpha_{k_3=2}$	.950	.932	.964	.948
$\beta_{k_2=1\|c_1^y=1}$	.952	.770	.962	.964
$\beta_{k_2=1\|c_1^y=2}$	.954	.880	.966	.962
$\beta_{k_2=2\|c_1^y=1}$	.960	.890	.964	.956
$\beta_{k_2=2\|c_1^y=2}$	.964	.884	.956	.954
$\beta_{k_3=1\|c_2^y=1}$	.944	.922	.940	.950
$\beta_{k_3=1\|c_2^y=2}$	.952	.936	.966	.968
$\beta_{k_3=2\|c_2^y=1}$	.974	.948	.952	.954
$\beta_{k_3=2\|c_2^y=2}$	.950	.860	.960	.966

Notes. LTA= latent transition analysis; MNAR-PP LTA=Missing-not-at-random parallel process LTA; MAR=missing-at-random.

Table OA11. Simulation results: Coverage for threshold measurement parameters in the outcome process

Parameter	MNAR Missingness mechanism		MAR Missingness mechanism	
	Fit MNAR-PP LTA	Fit Conventional LTA	Fit MNAR-PP LTA	Fit Conventional LTA
$v_{y1t\|k_t=1}$	.944	.942	.946	.944
$v_{y2t\|k_t=1}$	.940	.968	.942	.954
$v_{y3t\|k_t=1}$	.976	.988	.946	.952
$v_{y4t\|k_t=1}$	.940	.950	.952	.954
$v_{y5t\|k_t=1}$	.944	.952	.954	.954
$v_{y1t\|k_t=2}$	.932	.916	.936	.938
$v_{y2t\|k_t=2}$	.930	.914	.930	.934
$v_{y3t\|k_t=2}$	.932	.936	.920	.918
$v_{y4t\|k_t=2}$	.944	.908	.924	.932
$v_{y5t\|k_t=2}$	.944	.926	.928	.922
$v_{y1t\|k_t=3}$	.968	.920	.968	.970
$v_{y2t\|k_t=3}$	.936	.932	.934	.938
$v_{y3t\|k_t=3}$	.946	.918	.962	.962
$v_{y4t\|k_t=3}$	.944	.926	.962	.958
$v_{y5t\|k_t=3}$	.956	.912	.966	.970

Notes. LTA= latent transition analysis; MNAR-PP LTA=missing-not-at-random parallel process LTA; MAR=missing-at-random.

Table OA12. Simulation results: Coverage for multinomial coefficient structural parameters and threshold measurement parameters in the missingness process (available only when MNAR-PP LTA is fit)

	MNAR Missingness Mechanism	MAR Missingness Mechanism
Parameter	Fit MNAR-PP LTA	Fit MNAR-PP LTA
$\alpha_{q_2=1}$	.958	.956
$\alpha_{q_3=1}$	.970	.950
$\beta_{q_2=1\|c_1^y=1}$	.952	.952
$\beta_{q_2=1\|c_1^y=2}$	.960	.944
$\beta_{q_2=1\|c_2^y=1}$	.956	.962
$\beta_{q_2=1\|c_2^y=2}$	.958	.952
$\beta_{q_3=1\|c_2^m=1}$	.960	.944
$\beta_{q_3=1\|c_2^y=1}$	.960	.946
$\beta_{q_3=1\|c_2^y=2}$	.960	.960
$\beta_{q_3=1\|c_3^y=1}$	.966	.948
$\beta_{q_3=1\|c_3^y=2}$	.976	.960
$\nu_{m1t\|q_t=1}$	.936	.972
$\nu_{m2t\|q_t=1}$	.952	.938
$\nu_{m3t\|q_t=1}$	.952	.956
$\nu_{m4t\|q_t=1}$	.954	.946
$\nu_{m5t\|q_t=1}$	.954	.946
$\nu_{m1t\|q_t=2}$	.938	.946
$\nu_{m2t\|q_t=2}$	.940	.944
$\nu_{m3t\|q_t=2}$	.958	.940
$\nu_{m4t\|q_t=2}$	.930	.948
$\nu_{m5t\|q_t=2}$	.964	.954

Notes. MNAR-PP LTA=missing-not-at-random parallel process LTA; MAR=missing-at-random.

M*plus* 7.11 Syntax for MNAR-SP LTA with missingness starting at time 1 (as in Figure 4) where *J*=8, *K*=*Q,* and *T*=3 (See Manuscript Section 6.2 for special limitations of this model).

```
    ! comments only provided for commands that differ from previous syntax
DATA:  FILE = yourdataset2.dat;
VARIABLE: NAMES =  id t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t1m1-t1m8 t2m1-t2m8 t3m1-t3m8;
MISSING= . ;
USEVARIABLES ARE t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t1m1-t1m8 t2m1-t2m8 t3m1-t3m8;
CATEGORICAL = t1y1-t1y8 t2y1-t2y8 t3y1-t3y8 t1m1-t1m8 t2m1-t2m8 t3m1-t3m8;
CLASSES = c1 (4) c2 (4) c3 (4) ;
    ! now single process model has just one categorical latent variable at each t1-t3 (labeled c1-c3)
    ! now, for single process model, specify just one # of latent states per timepoint (here, 4)
ANALYSIS: TYPE = MIXTURE; STARTS=50 5; ESTIMATOR=ML;

MODEL:
%OVERALL% !now structural relations reflect constraints in Section 6.2 & Figure 4
c2 on c1;
c3 on c2;
    ! below, measurement invariance imposed within-state across times 1-3
    ! via list constraint (a1-a16) for thresholds in state 1, (b1-b16) for thresholds in state 2,
    ! (c1-c16) for thresholds in state 3,and (d1-d16) for thresholds in state 4
MODEL c1: !submodel at time 1 with thresholds for J=8 y-outcomes & J=8 m-indicators
%c1#1%  [t1y1$1-t1y8$1 t1m1$1-t1m8$1] (a1-a16);
%c1#2%  [t1y1$1-t1y8$1 t1m1$1-t1m8$1] (b1-b16);
%c1#3%  [t1y1$1-t1y8$1 t1m1$1-t1m8$1] (c1-c16);
%c1#4%  [t1y1$1-t1y8$1 t1m1$1-t1m8$1] (d1-d16);
MODEL c2y: !submodel at time 2 with thresholds for J=8 y-outcomes & J=8 m-indicators
%c2#1%  [t2y1$1-t2y8$1 t2m1$1-t2m8$1] (a1-a16);
%c2#2%  [t2y1$1-t2y8$1 t2m1$1-t2m8$1] (b1-b16);
%c2#3%  [t2y1$1-t2y8$1 t2m1$1-t2m8$1] (c1-c16);
%c2#4%  [t2y1$1-t2y8$1 t2m1$1-t2m8$1] (d1-d16);
MODEL c3y: !submodel at time 3 with thresholds for J=8 y-outcomes & J=8 m-indicators
%c3#1%  [t3y1$1-t3y8$1 t3m1$1-t3m8$1] (a1-a16);
%c3#2%  [t3y1$1-t3y8$1 t3m1$1-t3m8$1] (b1-b16);
%c3#3%  [t3y1$1-t3y8$1 t3m1$1-t3m8$1] (c1-c16);
%c3#4%  [t3y1$1-t3y8$1 t3m1$1-t3m8$1] (d1-d16);
```

Note. MNAR-SP LTA= Missing not at random single process latent transition model. Missingness indicators t1m1-t1m8, t2m1-t2m8, and t3m1-t3m8 are 1 if missing, 0 if present.